```
(* We verify Gauss' Theorem for the cylinder
 with radius R standing on top of the origin *)

In[159]:=
F = {x + z, y, z^2}
R = 3
H = 2

Out[159]= {x + z, y, z^2}

Out[160]= 3

Out[161]= 2

In[162]:= (* parametrization of side *)
r = {R Cos[u], R Sin[u], v}
ru = D[r, u]
rv = D[r, v]
n = Simplify[Cross[ru, rv]]

Out[162]= {3 Cos[u], 3 Sin[u], v}

Out[163]= {-3 Sin[u], 3 Cos[u], 0}

Out[164]= {0, 0, 1}

Out[165]= {3 Cos[u], 3 Sin[u], 0}

In[166]:= (* Flux through side *)
FOr = F /. {x → r[[1]], y → r[[2]], z → r[[3]]}
Simplify[Dot[FOr, n]]
Integrate[%, {u, 0, 2 Pi}]
I1 = Integrate[%, {v, 0, H}]

Out[166]= {v + 3 Cos[u], 3 Sin[u], v^2}

Out[167]= 9 + 3 v Cos[u]

Out[168]= 18 π

Out[169]= 36 π

In[170]:= (* parametrization of top *)
r = {v Cos[u], v Sin[u], H}
ru = D[r, u]
rv = D[r, v]
n = Simplify[Cross[rv, ru]]

Out[170]= {v Cos[u], v Sin[u], 2}

Out[171]= {-v Sin[u], v Cos[u], 0}

Out[172]= {Cos[u], Sin[u], 0}

Out[173]= {0, 0, v}
```

In[174]:=
```
(* Flux through top *)
FOr = F /. {x → r[[1]], y → r[[2]], z → r[[3]]}
Dot[FOr, n]
Integrate[%, {u, 0, 2 Pi}]
I2 = Integrate[%, {v, 0, R}]
```

Out[174]= {2 + v Cos[u], v Sin[u], 4}

Out[175]= 4 v

Out[176]= 8 π v

Out[177]= 36 π

In[178]:=
```
(* parametrization of bottom *)
r = {v Cos[u], v Sin[u], 0}
ru = D[r, u]
rv = D[r, v]
n = Simplify[Cross[ru, rv]]
```

Out[178]= {v Cos[u], v Sin[u], 0}

Out[179]= {-v Sin[u], v Cos[u], 0}

Out[180]= {Cos[u], Sin[u], 0}

Out[181]= {0, 0, -v}

In[185]:=
```
(* Flux through bottom *)
FOr = F /. {x → r[[1]], y → r[[2]], z → r[[3]]}
Dot[FOr, n]
Integrate[%, {u, 0, 2 Pi}]
I3 = Integrate[%, {v, 0, R}]
```

Out[185]= {v Cos[u], v Sin[u], 0}

Out[186]= 0

Out[187]= 0

Out[188]= 0

In[189]:=
```
(* Total flux *)
I1 + I2 + I3
```

Out[189]= 72 π

In[244]:=

```
(* Triple integral using cylindrical coordinates *)
r = {v Cos[u], v Sin[u], w}
rp = D[r, {{u, v, w}}]; MatrixForm[rp]
Jac = Abs[Simplify[Det[rp], a]]
dF = Div[F, {x, y, z}]
dFOr = dF /. {x → r[[1]], y → r[[2]], z -> r[[3]]}
Integrate[dFOr Jac, {u, 0, 2 Pi}]
Integrate[%, {v, 0, R}]
Integrate[%, {w, 0, H}]
```

Out[244]= {v Cos[u], v Sin[u], w}

Out[245]//MatrixForm=

$$\begin{pmatrix} -v\,\mathrm{Sin}[u] & \mathrm{Cos}[u] & 0 \\ v\,\mathrm{Cos}[u] & \mathrm{Sin}[u] & 0 \\ 0 & 0 & 1 \end{pmatrix}$$

Out[246]= Abs[v]

Out[247]= 2 + 2 z

Out[248]= 2 + 2 w

Out[249]= 2 π (2 + 2 w) Abs[v]

Out[250]= 18 (π + π w)

Out[251]= 72 π